Operaciones, Matrices y Polinomios

1) Realice las siguientes operaciones:

a) $\dfrac{\sqrt{34\cdot e^{2}}}{\cos(23.7)+12}$

b) $7\cdot e^{\frac{4}{5}}+3.54$

c) $20.\log\left(\dfrac{25.4}{\pi}\right)$

d) $(3+5i)^{3}-\sqrt{\dfrac{1}{(0.5-3i)}}$

```
//Ejercicio n°1
y=sqrt(34*%e^2)/(cos(23.7)+12)
y=7*exp(4/5)+3.54
y=20*log10(25.4/%pi)
y=(3+5*%i)^3-sqrt(1/(0.5-3*%i))
```

2) Dadas las siguientes variables: a=133, b=15, c=3.3x10^{-4}; calcule:

a) El resto de dividir a sobre b.

b) La raíz cuadrada de un número complejo cuya parte real sea -b y cuya parte imaginaria sea el log(a).

c) El módulo del resultado del inciso anterior multiplicado por c.

```
//Ejercicio n°2
a=133;b=15;c=3.3e-4;
x=modulo(a,b)
x=sqrt(-b+log10(a)*%i)
y=abs(c*x)
```

3) Dada la matrices $A=\begin{bmatrix}1 & -1 & 1 & 0\\ 2 & 4 & 0 & 1\\ 3 & 1 & 1 & 2\end{bmatrix}$ y $B=\begin{bmatrix}1 & 2 & 3 & 4\\ 2 & 1 & 4 & 1\\ 3 & 2 & 1 & 2\\ 4 & 3 & 2 & 1\\ 5 & 1 & 2 & 2\end{bmatrix}$

a) Para la matriz B extraer los bloques formados por:
   1. La fila cuarta
   2. La columna tercera
   3. Desde la fila 2 a la 4
   4. Desde la columna 1 a la columna 3
   5. Las filas 1,3 y 5
   6. Las columnas 2 y 4

b) Calcule el producto de la matriz A por una matriz formada por las filas 1,4,5 y 2 de la matriz B(en ese orden).

c) Determine las dimensiones de la matriz A.

d) Agregue una columna de unos a la matriz B y calcule su inversa.

e) Calcule el determinante de la matriz que resulta de eliminar la tercera columna de la matriz A. Calcule el rango.

```
//Ejercicio n°3
clear
clc
A=[1,-1,1,0;2,4,0,1;3,1,1,2]
B=[1,2,3,4;2,1,4,1;3,2,1,2;4,3,2,1;5,1,2,2]
// a)
C=B(4,:)
C=B(:,3)
C=B(2:4,:)
C=B(:,1:3)
C=B(1:2:5,:)
C=B(:,2:2:4)
// b)
C=A*[B(1,:);B(4:5,:);B(2,:)]
// c)
size(A)
// d)
C=inv([B,ones(5,1)])
// e)
C=A
C(:,3)=[]
det(C)
rank(C)
```

4) Construya las siguientes matrices:

a) Matriz identidad de orden 4
b) Matriz de 3x4 de valores aleatorios comprendidos entre -10 y 10.
c) Matriz de ceros de 4x4.
d) Matriz diagonal de 6x6, en la cual los elementos de la diagonal se encuentran equi-espaciados en el intervalo (-6,6).

```
//Ejercicio n°4
clear
clc
A=eye(4,4)
B=10-20*rand(3,4)
C=zeros(4,4)
D=diag(linspace(-6,6,6))
```

5) Evaluación de las operaciones de matrices con punto.

Genere 2 matrices (A y B) aleatorias de 4x4.

Realice las siguientes operaciones y analice sus diferencias: A*B ; A.*B ; A^2 ; A.^2 ; A/B ; A\B ; A./B y A.\B.

```
//Ejercicio n°5
clear
clc
A=5*rand(4,4)
B=0.2-0.4*rand(4,4)
C=A*B
C=A.*B
C=A^2
C=A.^2
C=A/B
C=A\B
C=A./B
C=A.\B
```

6) Calcule el polinomio característico y sus correspondientes raíces para la siguiente matriz:

$$A = \begin{bmatrix} -33 & -292 & -660 & -400 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \end{bmatrix}$$

Calcule el valor que toma el polinomio en sus raíces.

7) Construya un polinomio cuyas raíces sean: (0 , -1+i , -1-i)

8) Calcule el cociente entre los polinomios del ejercicio 6) y del ejercicio 7) (6/7). Verifique el resultado.

```
//Ejercicio n°6
clear
clc

A=[-33.  -292.  -660.  -400.
   1.    0.     0.     0.
   0.    1.     0.     0.
   0.    0.     1.     0.   ]
P=poly(A,"s")
R=roots(P)
horner(P,R)

//Ejercicio n°7

Q=poly([0,-1+%i,-1-%i],"s","roots")

//Ejercicio n°8

[resto,cociente]=pdiv(P,Q)

T=cociente*Q+resto
```

Gráficos

9) Grafique el polinomio del ejercicio 6) y verifique sus raíces.

```
//Ejercicio n°9
clc
t=-30:0.01:10;
y=horner(P,t);
plot2d(t,y);
xgrid()
```



10) Graficar las siguientes funciones:

a) $f(x) = \sqrt{x^2 - 1}$ en el dominio $x = [-3, 3]$.

```
//Ejercicio n°10
//a
x=-3:0.01:3;
y=sqrt(x^2-1);
plot2d(x,y);
xgrid()
```



b) $f(x)=\sqrt{1-(|x|-1)^2}$ y $g(x)=\arccos(1-|x|)-\pi$ en el dominio $x=[-2,2]$.

```
//b
clear
clc
x=-2:0.05:2;
f=sqrt(1-(abs(x)-1)^2)
g=acos(1-abs(x))-%pi
plot2d(x,f);
plot2d(x,g);
xgrid()
```



c) $f(x,y)=\dfrac{seno\left(\sqrt{x^2+y^2}\right)}{\sqrt{x^2+y^2+0.1}}$ en el dominio $x=[-10,10]$ e $y=[-10,10]$.

```
//c
clear
clc
u=-10:0.5:10;
v=-10:0.5:10;
for i=1:length(u)
    for j=1:length(v)
        z(i,j)=100*sin(sqrt(u(i)^2+v(j)^2))/(sqrt(u(i)^2+v(j)^2+0.1));
    end
end
plot3d(u,v,z)
xgrid()
```



Funciones

11) Programe una función en la cual ingresando un número complejo**,** calcula el módulo en dB (20*log(módulo)) y la fase en grados.

```
//Ejercicio n°11
clear
clc

function [dB, Fase]=Modfi(complejo)
    // esta función calcula el módulo en db de un número complejo
    // y su fase en grados sexagesimales.
    modulo=sqrt(real(complejo)^2+imag(complejo)^2);
    dB=20*log10(modulo);
    Fase=180/%pi*atan(imag(complejo)/real(complejo));
endfunction

[m,f]=Modfi(100-100*%i)
```

12) Programar una función que permita calcular el valor de las siguientes representaciones por tramos:

$$f(x)=\begin{cases} x+2 & si\ x\le -1 \\ -x & si -1<x\le 1 \\ 3x-4 & si\ \ x>1 \end{cases}$$

```
//Ejercicio n°12a
clear
clc

function y=Tramoa(x)
    // esta función calcula el valor de una función de x por tramos
    // x es un vector de n componentes
    for i=1:length(x)
        valor=x(i)
        if valor<=-1 then
            y(i)=valor+2;
        elseif valor>-1 & valor<=1
            y(i)=-valor
        else
            y(i)=3*valor-4
        end
    end
endfunction

m=Tramoa(-5)
n=Tramoa(0)
p=Tramoa(5)

t=linspace(-10,10,100);
y1=Tramoa(t);
plot2d(t,y1);
xgrid();
```



$$f(x)=\begin{cases} x^2-3x & \text{si } x\le 0 \\ x & \text{si } 0<x\le 2 \\ 2x-2 & \text{si }\ x>2 \end{cases}$$

```
//Ejercicio n°12b
clear
clc
function y=Tramob(x)
    // esta función calcula el valor de una función de x por tramos
    // x es un vector de n componentes
    for i=1:length(x)
        valor=x(i)
        if valor<=0 then
            y(i)=valor^2-3*valor;
        elseif valor>0 & valor<=2
            y(i)=valor
        else
            y(i)=2*valor-2
        end
    end
endfunction

m=Tramob(-5)
n=Tramob(0)

t=linspace(-4,20,100);
y1=Tramob(t);
plot2d(t,y1);
xgrid();
```